

1 Start and Menu

Launching the Fraud Toolkit Tests


Launching the `_Start` script leads to the Fraud Menu, which, in turn, is the gateway to the 36 ACL scripts that make up the Fraud Toolkit. The scripts are linked through a hierarchical menu structure.

The `_Start` and `_Fraud.Menu` scripts allow the user to easily perform a variety of fraud tests. After the user selects tests from the Fraud Menu, the scripts that comprise the tests run, prompting the user for the required input through a series of easy-to-use dialog boxes. The results of these tests either appear in a dialog box or are written to an output file. In addition, the results of all tests are written to the command log. The scripts are part of an ACL project file named “Fraud Toolkit.” To begin using the scripts, the user must first open the Fraud Toolkit project, or the working project to which the scripts have been copied. See “Preparing Scripts for Use” on page 7 for detailed instructions on installing and using the Fraud Toolkit scripts. Next run the `_Start` script to open the Fraud Menu and view the available tests.


Starting the Fraud Toolkit Application

To launch the `_Start` script and display the Fraud Menu:

1. In the Project Navigator window, click the plus + (Expand) next to the Scripts subfolder or the Scripts icon to expand the list of scripts.
2. Right-click `_Start`. ACL will ask if you want to run or edit the script. Click [Run]. The Welcome dialog box appears:



3. Click [OK] to display the Fraud Menu dialog box; then select the check box for each group of tests to be run. The tests will run in the order in which they appear on the menu. To return to the menu after the tests are complete, check the appropriate box:



_Start

```

SET ECHO NONE
ASSIGN fraud=T
ASSIGN exact_off=F
IF 'A'='ABC' exact_off=T
SET EXACT ON
DIALOG (DIALOG TITLE "Fraud Toolkit for ACL"
  WIDTH 444 HEIGHT 325 ) (BUTTONSET TITLE "OK"
  AT 168 228 DEFAULT 1 HORZ ) (TEXT TITLE
  "Press Enter or click OK to continue" AT
  120 196 WIDTH 198 ) (TEXT TITLE "WELCOME" AT
  180 40) (TEXT TITLE "Fraud Toolkit for ACL"
  AT 156 88 HEIGHT 18 ) (TEXT TITLE "Copyright
  2009 David G. Coderre" AT 300 244 WIDTH 94
  HEIGHT 33 ) (TEXT TITLE "Fraud Analysis
  Techniques Using ACL" AT 120 124 WIDTH 195 )
  (TEXT TITLE "Ver 1.0" AT 348 280 HEIGHT 23 )
  (TEXT TITLE "John Wiley and Sons" AT 156 160
  WIDTH 110 )

DO _Fraud_Menu WHILE fraud
IF exact_off SET EXACT OFF
DELETE ALL OK
SET ECHO ON

```

Note: By default, a log file with the same name as the ACL Project file will be created when the Project is created. An alternate log file can be specified by typing the log file name, for example, *Payroll Fraud Tests*, in the Specify log file text box. This creates a log file called *Payroll_Fraud_Tests.log*, or, if that file already exists, appends the new results to the existing log file.

Placement of Start and Fraud Menu Scripts

The Start script is preceded by a double underscore (`_Start`), and the Fraud Menu script is preceded by a single underscore (`_Fraud_Menu`). This ensures that these two key scripts are easily found at the top of the Project Navigator window, and that `_Start` appears before `_Fraud_Menu`:

How the Scripts Work

A menu-based application such as the Fraud Toolkit has a number of advantages. It conveniently organizes all the available tests and allows the user to run several tests consecutively. If used correctly, the menu also ensures that, once all testing is complete, any temporary variables, files, or settings are properly deleted or restored.

Start

The two most important functions of the Start script are activating the system condition Exact Character Comparisons (**SET EXACT ON**) and creating a variable, **fraud**, that tests whether the Fraud Menu dialog box is still needed.

The Start script checks whether it should reset the ACL system condition to **SET EXACT OFF** when done. If the user decides to exit the Fraud Menu, the **fraud** variable is set to False, which ends the `_Fraud_Menu` script and returns processing to the `_Start` script. As evident in the code, the `_Start` script then deletes the temporary variables that are no longer needed, and, if necessary,

resets system conditions. This efficient programming device is employed several times in the Toolkit scripts. The `_Start` script also includes a dialog box that introduces the application and displays important source and copyright information.

Note: Be sure to read the copyright page before modifying these scripts for use in your own organization.

Fraud Menu

The `_Fraud_Menu` script gives the user the option of selecting, from a series of check boxes, one or more tests to run. Notice that the default is to exit this menu once the tests have run. To return to the Fraud Menu, select the appropriate check box.

Modifying the Fraud Menu is easy. Adding or removing menu choices is as simple as changing two lines in the script. For example, suppose the user wanted to add an option to run a script to generate confirmation letters or review user authorizations. Simply modify the dialog box in the Fraud Menu script by adding a check box for this purpose, and then add another `If . . . Do` statement to launch the new test whenever that check box is selected:

Log Files

The `_Fraud_Menu` script can also be used to create a different log file to store the contents of the command log. This optional feature is extremely useful for occasions when it is important to separate the results from non fraud-related analyses that were performed. For example, the user might want to keep the results of an audit-related analysis separate from the results of a fraud investigation. By default, the results of all analyses are written to a log file with the same name as the ACL project. By entering a different file name when running the Fraud Toolkit, the user can direct the results to a file name of their choice. In cases where the person committing the fraud is prosecuted,

Fraud_Menu

```
SET ECHO NONE
DIALOG (DIALOG TITLE "Fraud Menu" WIDTH 740
HEIGHT 400 ) (BUTTONSET TITLE "&OK;&Cancel"
AT 612 24 DEFAULT 1 ) (TEXT TITLE "Select
one or more fraud tests:" AT 72 40 WIDTH 202
HEIGHT 22 ) (TEXT TITLE "Specify log file
(optional)" AT 72 352 ) (EDIT TO "log_file"
AT 240 348 WIDTH 317 ) (CHECKBOX TITLE
"Completeness and Integrity" TO "menu1" AT
132 72 WIDTH 230 ) (CHECKBOX TITLE "Cross-
Tabulation" TO "menu2" AT 132 96 ) (CHECKBOX
TITLE "Duplicates" TO "menu3" AT 132 120 )
(CHECKBOX TITLE "Gaps" TO "menu4" AT 132
144 ) (CHECKBOX TITLE "Data Profile" TO
"menu5" AT 132 168 ) (CHECKBOX TITLE "Ratio
Analysis" TO "menu6" AT 132 192 ) (CHECKBOX
TITLE "Benford Analysis" TO "menu7" AT 132
216 ) (CHECKBOX TITLE "Return to this menu
when done" TO "menu8" AT 132 264 )

logout=REPLACE('%log_file%',' ','_')
IF LEN(exclude(logout,'_'))>0 SET LOG TO
%logout%
SET ECHO NONE
IF menu1 DO Complete
SET ECHO NONE
IF menu2 DO Cross_Tabs
SET ECHO NONE
IF menu3 DO Dup_Multiple_Keys
SET ECHO NONE
IF menu4 DO Gaps
SET ECHO NONE
IF menu5 DO Data_Profile
```

```
SET ECHO NONE
IF menu6 DO Ratio_Analysis
SET ECHO NONE
IF menu7 DO Benford
SET ECHO NONE
IF NOT menu8 fraud=F
SET ECHO NONE
DELETE log_file OK
DELETE logout OK
DELETE menu1 OK
DELETE menu2 OK
DELETE menu3 OK
DELETE menu4 OK
DELETE menu5 OK
DELETE menu6 OK
DELETE menu7 OK
DELETE menu8 OK
SET LOG
SET ECHO ON
```

and you are required to provide evidence, having a separate log file with only the results of the fraud analyses can be highly beneficial.

To create an alternate log file, the script performs these steps:

- The Dialog command prompts the user for a command log name, and then captures and assigns the user input to a character variable named `log_file`.
- The script modifies the user-supplied name to eliminate any spaces (ACL file names cannot have spaces). The `REPLACE()` function is used to change all spaces to underscores in whatever character string was typed in the dialog box:

```
logout= replace ('%log_file%', ' ', '_')
```

The script tests to see if the length of the modified name is greater than zero characters. If the test result is true, the command log results are temporarily redirected to the specified alternate log file. When the fraud tests have finished running and you have exited the Fraud Menu, any subsequent results are sent to the default log file.

To review the results, open the alternate log file by using the Set Log command:

```
SET LOG TO log_file name
```

Then, to set the log back to the default log file, specify:

```
SET LOG OFF
```

If the application will be run several times during an investigation, the user can specify the alternate log name each time he or she selects tests from the Fraud Menu. The alternate log will then be opened and the results will be appended. The log can be set to an alternate file *before* running the fraud application, in which case there is no need for further action. The new file is the default log file throughout the session.

Exiting a Script

It is possible to exit a script that was launched from the menu by clicking either the **Cancel** button or the **Close** button (top, right corner of every dialog box). This will be necessary if, for example, a script is launched by mistake and you do not wish to run it. However, exiting by this method is not recommended because it interrupts the normal processing of the script application. When a script is interrupted, temporary variables, fields, or files may not be properly deleted, and other system conditions like **ECHO ON** may not be restored.

Note: If a script is interrupted, check the project folder for added files, check the table layouts for computed fields that were not deleted, and remove temporary variables by typing `DELETE ALL OK` on the command line.

Working without the Fraud Menu


All instructions given in this book assume that the user will run tests using the Fraud Menu. However, experienced users may choose to bypass the menu and run certain scripts separately by selecting them from the Project Navigator window. The scripts for the main tests will work if run individually; they each contain commands to delete any temporary variables, fields, and files that they create. The supporting scripts, called by the main test scripts, can also be run individually, but most do not perform tests when run alone. For example, Complete.Dialog collects input from the user, but if the Complete script is not run first, the collected information will not be processed.

When working without the Fraud Menu, the user must ensure that the **Exact Character Comparisons** option is selected in ACL

Preferences. If this option is not selected, tests that depend on an exact comparison may produce different results or may not run at all. If, for example, two fields, Amount and Amount2, are selected with the preference turned off, there is a risk that ACL will treat them both as the same field and only analyze the field named Amount.

If the Fraud ToolKit is run from the `_Start` script, the user doesn't have to be concerned about this option; the `_Start` script automatically activates Exact Character Comparisons. However, if the individual scripts are run by double-clicking them, or if the user builds his or her own application, this option should be selected. Set this preference as follows:

1. From the **Tools** menu select **Options**
2. On the Table tab select the **Exact Character Comparisons** check box:



The following chart depicts the relationships of the various scripts.

